Pensieve header: A free-Lie calculator, at long last. (Continues pensieve://2011-02/).

A Lyndon word is a word lexicographically smaller than all of its proper right factors; see http://katlas.math.toronto.edu/drorbn/bbs/show?shot=Chu-071214-182203.jpg

```
LyndonQ[w_String] := And @@ (
    OrderedQ[{w, #}] & /@ Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]
   );
AllWords[0, _List] = {""};
AllWords[n_ /; n > 0, ab_List] := AllWords[n, ab] = Flatten[Outer[
     StringJoin[#1, #2] &,
     AllWords[n - 1, ab],
     ab
    ]];
AllLyndonWords[n_Integer, ab_List] := LW /@ Select[AllWords[n, ab], LyndonQ];
AllLyndonWords[{n_}, ab_List] := Join @@ Table[AllLyndonWords[k, ab], {k, n}];
LyndonFactorization[LW[w_String] /; StringLength[w] == 1] := LW[w];
LyndonFactorization[LW[w_String] /; StringLength[w] > 1] := Module[
   {rf},
   rf = First[Sort[Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]]];
   LW /@ {StringDrop[w, -StringLength[rf]], rf}
  ];
LW /: LW[x_] ≤ LW[y_] := OrderedQ[{x, y}];
LW /: x_LW ≥ y_LW := y ≤ x;
LW /: x_LW > y_LW := ! (x ≤ y);
LW /: x_LW < y_LW := ! (y ≤ x);
Format[LW[w_], StandardForm] := Defer[〈w〉];
〈w_〉 := LW[w];
Deg[LW[x_]] := StringLength[x];
```

```
LyndonQ["abba"]
```

False

```
LyndonQ["ababb"]
```

True

```
AllWords[1, {"1", "2"}]
```

{1, 2}

```
AllWords[3, {"1", "2"}]
```

{111, 112, 121, 122, 211, 212, 221, 222}

```
AllLyndonWords[3, {"1", "2"}]
```

{〈112〉, 〈122〉}

```
Table[Length[AllLyndonWords[k, {"1", "2"}]], {k, 10}]
```

{2, 1, 2, 3, 6, 9, 18, 30, 56, 99}

```
Table[Length[AllLyndonWords[k, {"1", "2", "3"}]], {k, 8}]
```

{3, 3, 8, 18, 48, 116, 312, 810}

```
TreeForm[LW["12122"] //. w_LW :→ LyndonFactorization[w] /. LW[w_] :→ w]
```




```
b[0, _] = 0; b[_, 0] = 0;
b[c_ * x_LW, y_] := Expand[c b[x, y]];
b[x_, c_ * y_LW] := Expand[c b[x, y]];
b[x_Plus, y_] := b[#, y] & /@ x;
b[x_, y_Plus] := b[x, #] & /@ y;
b[w_LW, z_LW] := Which[
   w === z, 0,
   z < w, Expand[-b[z, w]],
   Deg[w] == 1, LW[First[w] <> First[z]],
   True, Module[{x, y},
    {x, y} = LyndonFactorization[w];
    If[y ≥ z,
     LW[First[w] <> First[z]],
     b[x, b[y, z]] + b[b[x, z], y]
    ]
   ]
  ];
```

```
b[LW["112"], LW["122"]]
```

```
⟨112122⟩ + ⟨112212⟩
```

```
Outer[b, AllLyndonWords[5, {"1", "2"}], AllLyndonWords[5, {"1", "2"}]]
```

```
{{0, ⟨1111211122⟩ + ⟨1111221112⟩,
  ⟨1111211212⟩ + ⟨1111212112⟩ - ⟨1112111212⟩, ⟨1111211222⟩ + ⟨1111222112⟩ - ⟨1112111222⟩,
  ⟨1111212122⟩ + ⟨1111212212⟩ - 2 ⟨1112112122⟩ - ⟨1112122112⟩,
  ⟨1111212222⟩ + ⟨1111222212⟩ - 2 ⟨1112112222⟩ - ⟨1112222112⟩},
 {-⟨1111211122⟩ - ⟨1111221112⟩, 0, ⟨1112211212⟩, ⟨1112211222⟩ + ⟨1112221122⟩,
  ⟨1112212122⟩ + ⟨1121221122⟩, ⟨1112212222⟩ + ⟨1112222122⟩ - ⟨1122112222⟩},
 {-⟨1111211212⟩ - ⟨1111212112⟩ + ⟨1112111212⟩, -⟨1112211212⟩, 0,
  ⟨1121211222⟩, ⟨1121212122⟩ + 2 ⟨1121212212⟩ + ⟨1121221212⟩,
  ⟨1121212222⟩ + 2 ⟨1121222212⟩ + ⟨1122221212⟩},
 {-⟨1111211222⟩ - ⟨1111222112⟩ + ⟨1112111222⟩, -⟨1112211222⟩ - ⟨1112221122⟩,
  -⟨1121211222⟩, 0, ⟨1122212122⟩, ⟨1122212222⟩ + ⟨1122221222⟩},
 {-⟨1111212122⟩ - ⟨1111212212⟩ + 2 ⟨1112112122⟩ + ⟨1112122112⟩,
  -⟨1112212122⟩ - ⟨1121221122⟩, -⟨1121212122⟩ - 2 ⟨1121212212⟩ - ⟨1121221212⟩,
  -⟨1122212122⟩, 0, ⟨1212212222⟩ + ⟨1212222122⟩},
 {-⟨1111212222⟩ - ⟨1111222212⟩ + 2 ⟨1112112222⟩ + ⟨1112222112⟩, -⟨1112212222⟩ -
   ⟨1112222122⟩ + ⟨1122112222⟩, -⟨1121212222⟩ - 2 ⟨1121222212⟩ - ⟨1122221212⟩,
  -⟨1122212222⟩ - ⟨1122221222⟩, -⟨1212212222⟩ - ⟨1212222122⟩, 0}}
```

```
⟨"1122221222"⟩
```

```
⟨1122221222⟩
```

```
⟨"1122221222"⟩ // FullForm
```

```
LW["1122221222"]
```

```
Length /@ Flatten[
  Outer[b, AllLyndonWords[5, {"1", "2"}], AllLyndonWords[5, {"1", "2"}]]
 ]
```

```
{0, 2, 3, 3, 4, 4, 2, 0, 1, 2, 2, 3, 3, 2, 0, 1,
 3, 3, 3, 2, 2, 0, 1, 2, 4, 2, 3, 2, 0, 2, 4, 3, 3, 2, 2, 0}
```

```
Outer[(b[#1, #2] + b[#2, #1]) &,
  AllLyndonWords[{6}, {"1", "2"}], AllLyndonWords[{6}, {"1", "2"}]
 ] // MatrixForm
```

```
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

```
Outer[(b[#1, b[#2, #3]] + b[#2, b[#3, #1]] + b[#3, b[#1, #2]]) &,
   AllLyndonWords[{5}, {"1", "2"}],
   AllLyndonWords[{5}, {"1", "2"}], AllLyndonWords[{5}, {"1", "2"}]
  ] // Flatten //
 Union
```

```
{0}
```

```
{x = LW["x"], y = LW["y"]}
```

```
{⟨x⟩, ⟨y⟩}
```

```
ad[x_][y_] := b[x, y];
MakeLieSeries[expr_, d_] := LieSeries @@ Table[
    expr /. w_LW /; Deg[w] ≠ k → 0,
    {k, d}
   ];
LieSeries /: s1_LieSeries + s2_LieSeries := Module[
   {l = Min[Length /@ {s1, s2}]},
   LieSeries @@ (Take[List @@ s1, l] + Take[List @@ s2, l])
  ];
LieSeries /: c_ * s_LieSeries := Expand[c * #] & /@ s;
b[s1_LieSeries, s2_LieSeries] := Module[
   {d, k},
   LieSeries @@ Table[
     Sum[b[s1[[k]], s2[[d - k]]], {k, 1, d - 1}],
     {d, 1 + Min[Length /@ {s1, s2}]}
    ]
  ];
LieSeries /: EulerE[s_LieSeries] :=
  LieSeries @@ Expand[Range[Length[s]] * (List @@ s)];
OperatorSeries[f_, var_ → op_, d_][ψ_] := Module[
  {ser, as, ni, nf, t},
  ser = List @@ Series[f, {var, 0, d}];
  {as, ni, nf} = ser[[{3, 4, 5}]];
  t = Nest[op, ψ, ni];
  Expand[as.NestList[op, t, nf - ni - 1]]
 ]
```

```
OperatorSeries[E^(-ad), ad → ad[y], 3][x]
```

$$\langle x\rangle + \langle xy\rangle + \frac{\langle xyy\rangle}{2} + \frac{\langle xyyy\rangle}{6}$$

```
MakeLieSeries[OperatorSeries[E^(-ad), ad → ad[x], 3][y], 5]
```

$$\text{LieSeries}\left[\langle y\rangle,\ -\langle xy\rangle,\ \frac{\langle xxy\rangle}{2},\ -\frac{\langle xxxy\rangle}{6},\ 0\right]$$

```
BCH[1] = LieSeries[〈"x"〉 + 〈"y"〉]
```

$\text{LieSeries}[\langle x\rangle + \langle y\rangle]$

```
BCH[n_] := Module[
  {bch, t1, t2},
  bch = Append[BCH[n - 1], 0];
  t1 =
   MakeLieSeries[〈"y"〉 + OperatorSeries[E^(-ad), ad → ad[〈"y"〉], n - 1][〈"x"〉], n];
  t2 = OperatorSeries[(1 - E^(-ad)) / ad, ad → ad[bch], n - 1][EulerE[bch]];
  bch + (t1 - t2) / n
 ]
```

```
t1 = MakeLieSeries[〈"y"〉 + OperatorSeries[E^(-ad), ad → ad[〈"y"〉], 1][〈"x"〉], 2]
```

$\text{LieSeries}[\langle x\rangle + \langle y\rangle,\ \langle xy\rangle]$

```
bch = Append[BCH[1], 0];
t2 = OperatorSeries[(1 - E^(-ad)) / ad, ad → ad[bch], 1][EulerE[bch]]
```

$$\text{LieSeries}[\langle x\rangle + \langle y\rangle,\ 0]$$

```
t1 - t2
```

$$\text{LieSeries}[0,\ \langle xy\rangle]$$

```
BCH[2]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2}\right]$$

```
BCH[3]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12}\right]$$

```
BCH[4]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24}\right]$$

```
BCH[5]
```

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},$$
$$-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720}\Big]$$

```
BCH[6]
```

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},$$
$$-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720},$$
$$-\frac{\langle xxxxyy\rangle}{1440} + \frac{\langle xxxyxy\rangle}{720} + \frac{\langle xxxyyy\rangle}{360} + \frac{\langle xxyxyy\rangle}{240} - \frac{\langle xxyyyy\rangle}{1440}\Big]$$

```
BCH[7]
```

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},$$
$$-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720},$$
$$-\frac{\langle xxxxyy\rangle}{1440} + \frac{\langle xxxyxy\rangle}{720} + \frac{\langle xxxyyy\rangle}{360} + \frac{\langle xxyxyy\rangle}{240} - \frac{\langle xxyyyy\rangle}{1440},$$
$$\frac{\langle xxxxxxy\rangle}{30240} - \frac{\langle xxxxxyy\rangle}{5040} + \frac{\langle xxxxyxy\rangle}{10080} + \frac{\langle xxxxyyy\rangle}{3780} + \frac{\langle xxxyxxy\rangle}{10080} + \frac{\langle xxxyxyy\rangle}{1680} +$$
$$\frac{\langle xxxyyxy\rangle}{1260} + \frac{\langle xxxyyyy\rangle}{3780} + \frac{\langle xxyxxyy\rangle}{2016} - \frac{\langle xxyxyxy\rangle}{5040} + \frac{13\,\langle xxyxyyy\rangle}{15120} + \frac{\langle xxyyxyy\rangle}{10080} -$$
$$\frac{\langle xxyyyxy\rangle}{1512} - \frac{\langle xxyyyyy\rangle}{5040} + \frac{\langle xyxyxyy\rangle}{1260} - \frac{\langle xyxyyyy\rangle}{2016} - \frac{\langle xyyxyyy\rangle}{5040} + \frac{\langle xyyyyyy\rangle}{30240}\Big]$$

BCH[**8**]

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},$$
$$-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720},$$
$$-\frac{\langle xxxxyy\rangle}{1440} + \frac{\langle xxxyxy\rangle}{720} + \frac{\langle xxxyyy\rangle}{360} + \frac{\langle xxyxyy\rangle}{240} - \frac{\langle xxyyyy\rangle}{1440},$$
$$\frac{\langle xxxxxxy\rangle}{30\,240} - \frac{\langle xxxxxyy\rangle}{5040} + \frac{\langle xxxxyxy\rangle}{10\,080} + \frac{\langle xxxxyyy\rangle}{3780} + \frac{\langle xxxyxxy\rangle}{10\,080} + \frac{\langle xxxyxyy\rangle}{1680} +$$
$$\frac{\langle xxxyyxy\rangle}{1260} + \frac{\langle xxxyyyy\rangle}{3780} + \frac{\langle xxyxxyy\rangle}{2016} - \frac{\langle xxyxyxy\rangle}{5040} + \frac{13\,\langle xxyxyyy\rangle}{15\,120} + \frac{\langle xxyyxyy\rangle}{10\,080} -$$
$$\frac{\langle xxyyyxy\rangle}{1512} - \frac{\langle xxyyyyy\rangle}{5040} + \frac{\langle xyxyxyy\rangle}{1260} - \frac{\langle xyxyyyy\rangle}{2016} - \frac{\langle xyyxyyy\rangle}{5040} + \frac{\langle xyyyyyy\rangle}{30\,240},$$
$$\frac{\langle xxxxxxyy\rangle}{60\,480} - \frac{\langle xxxxxyxy\rangle}{15\,120} - \frac{\langle xxxxxyyy\rangle}{10\,080} + \frac{\langle xxxxyxxy\rangle}{20\,160} - \frac{\langle xxxxyxyy\rangle}{20\,160} + \frac{\langle xxxxyyxy\rangle}{2520} +$$
$$\frac{23\,\langle xxxxyyyy\rangle}{120\,960} + \frac{\langle xxxyxxyy\rangle}{4032} - \frac{\langle xxxyxyxy\rangle}{10\,080} + \frac{13\,\langle xxxyxyyy\rangle}{30\,240} + \frac{\langle xxxyyxyy\rangle}{20\,160} -$$
$$\frac{\langle xxxyyyxy\rangle}{3024} - \frac{\langle xxxyyyyy\rangle}{10\,080} + \frac{\langle xxyxyxyy\rangle}{2520} - \frac{\langle xxyxyyyy\rangle}{4032} - \frac{\langle xxyyxyyy\rangle}{10\,080} + \frac{\langle xxyyyyyy\rangle}{60\,480}\Big]$$

BCH[**15**]

A very large output was generated. Here is a sample of it:

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\ll 1 \gg}{24},\ \ll 7 \gg,\ \ll 1 \gg,\ -\frac{691\,\langle xx \ldots xxy\rangle}{1\,307\,674\,368\,000} + \ll 948 \gg,$$
$$-\frac{691\,\langle xxxxxxxxxxxxyy\rangle}{2\,615\,348\,736\,000} + \frac{691\,\langle xxxxxxxxxxxyxy\rangle}{261\,534\,873\,600} + \ll 948 \gg + \frac{157\,\langle xxyyyyyxyyyyyy\rangle}{3\,459\,456\,000} - \frac{691\,\langle xxyyyyyyyyyyyy\rangle}{2\,615\,348\,736\,000},$$
$$\frac{\langle xxxxxxxxxxxxxxy\rangle}{74\,724\,249\,600} - \frac{\langle xxxxxxxxxxxxxyy\rangle}{5\,337\,446\,400} + \frac{\langle xxxxxxxxxxxxyxy\rangle}{1\,186\,099\,200} + \ll 3239 \gg + \frac{\langle xyyyyyyyyyyyyyy\rangle}{74\,724\,249\,600}\Big]$$

Show Less | Show More | Show Full Output | Set Size Limit...

TimeUsed[]

375.915